
BEE 271 Digital circuits and systems
Spring 2017

Lecture 11:  Sequential circuits

Nicole Hamilton
https://faculty.washington.edu/kd1uj

http://faculty.washington.edu/kd1uj


Topics

1. Introduction to sequential circuits



Terminology

Basic latch:  A feedback connection of two NORs or two 
NANDs that can store 1 bit.  Set using the S input and reset 
using the R input.

Gated latch:  A basic latch that includes input gating and a 
control input signal.  Retains its value when the control 
signal is 0, changes state when the control signal is 1.

Flip-flop:  A storage element whose output changes only on 
the edge of a controlling clock.  Can be either positive or 
negative edge-triggered.

Figure 5.3.   A memory element with NOR gates.

Figure 5.4.   A basic latch built with NOR gates.

module SetResetLatch( input set, reset,
output Q, Qn );

nor ( Qn, set,   Q  );
nor ( Q,  reset, Qn );

endmodule

module SetnResetnLatch( input setn, resetn,
output Q, Qn );

nand ( Q,  setn,   Qn );
nand ( Qn, resetn, Q  );

endmodule

module SRLatch( input s, r, output reg Q );

always @( * )
casex ( { s, r } )

'b1x: Q = 1;
'b01: Q = 0;

endcase

endmodule



Figure 5.6.   Gated SR latch with NAND gates.

module DLatch1( input clock, D, output reg Q );

always @( * )

if ( clock )

Q = D;     // Implied memory

endmodule

module DLatch2( input clock, D, output reg Q );

always @( * )

Q = clock ? D : Q;

endmodule



Figure 5.5.   
Gated SR latch.



Figure 5.7.   
Gated D 
latch.



Flip Flops

• Problem: latches are sensitive to any 
changes that occur with the input while 
the clock or control signal is high
– Glitches/Hazards
– Unsynchronized changes

• Solution: use flip-flops, devices that react 
only on the clock edge



But we cannot 
build reliable 
counters or 
anything else 
requiring 
feedback with 
latches.

What you get depends on small differences in the length of the green wire.



Solution is to isolate inputs from outputs

Three popular alternatives:
1. Two-phase clocking
2. Master-slave
3. Edge-triggered

Non-overlapping clocks for phases 1 and 2.  Advantages are fewer gates per bit and 
that you can put logic between both phases so long phase 1 latches only use phase 2 
inputs and vice versa.

At the positive edge, 
B  D and A  D*, 
disabling the input 
and copying A Q* 
and B  Q

A and B  1, 
latching Q

Requires a hold time:  D is not allowed to change until the rising edge of 

tsu + th is called the “aperture” or “window” when the input must be good.



Figure 5.9.   
Master-slave D 
flip-flop.

Figure 5.10.   Comparison 
of level-sensitive and 
edge-triggered D storage 
elements.

module DMasterSlave1( input clock, D, output reg Q );

wire Qm;
DLatch1 master ( clock,  D,  Qm );
DLatch1 slave  ( ~clock, Qm, Q  );

endmodule

module DMasterSlave2( input clock, D, output reg Q );

reg Qm;
always @( * )

if ( clock )
Qm = D;

else
Q = Qm;

endmodule



module DMasterSlave3( input clock, D, output reg Q );

reg Qm;
always @( negedge clock )

Q <= D;

endmodule



module DMasterSlave3( input clock, D, output reg Q );

reg Qm;
always @( negedge clock )

Q <= D;

endmodule

module DEdgeTriggered( input clock, D, output reg Q );

always @( posedge clock )
Q <= D;

endmodule



Figure 5.11.   A positive-edge-triggered D flip-flop.

D 

Clock 

P4

P3

P1

P2

5 

6 

1 

2 

3 

(a) Circuit 

D Q 

Q 

(b) Graphical symbol

Clock 

Q 

Q 

4 

Figure 5.7.  Gated D latch.

Steady-state clock = 0.

0

1

1

D’

D’
D

x

x’
The critical path through 
gates 4 and 1 determines 
the setup time.



Clock transition 0  1

D transitions 0  1 after the hold time

Figure 5.12.   Master-slave D flip-flop with Clear and Preset.



Figure 5.13.   
Positive-edge-
triggered D flip-flop 
with Clear and 
Preset.



Figure 5.14.  Timing for a flip-flop.



Figure 5.15.   T 
flip-flop.

Figure 5.16.   JK flip-flop.

Figure 5.17.   A simple shift register.

Figure 5.18.   Parallel-access shift register.



Basic D Flip Flop

D

Clk

Qff(posedge)

module DFlipFlop( input clock, D, output reg Q );

always @( posedge clock )
Q <= D;

endmodule



Basic D Flip Flop

module DFlipFlop( input clock, D, output reg Q );

always @( posedge clock )
Q <= D;

endmodule



D Flip Flop w/Synchronous Reset

module DFlipFlop2( input clock, D, reset, output reg Q );

always @( posedge clock )
Q <= reset ? 0 : D;

endmodule



Figure 5.19.   A three-bit up-counter.

Figure 5.20.   A three-bit down-counter.

Table 5.1.   Derivation of the synchronous up-counter.

Figure 5.21.   A four-bit synchronous up-counter.

Figure 5.22.   Inclusion of Enable and Clear capability.

Figure 5.23.   A four-
bit counter with D 
flip-flops.



Figure 5.24.   A 
counter with 
parallel-load 
capability.



Figure 5.25.   A modulo-6 counter with synchronous reset.

Figure 5.26.   A modulo-6 counter with asynchronous reset.

Figure 5.23.   A four-
bit counter with D 
flip-flops.



Figure 5.24.   A 
counter with 
parallel-load 
capability.



Figure 5.66.  A simple flip-flop circuit.

Assume:
tsu = 0.6 ns
th = 0.4 ns
0.8 ns <= tcQ <= 1.0 ns
tgate = 1.0 + 0.1k

where
k = number of inputs

Tmin = tcQmax +  tNOT + tsu

= 1.0 + 1.1 + 0.6 = 2.7 ns
Fmax = 1/2.7 ns = 370.37 MHz.



Figure 5.66.  A simple flip-flop circuit.

Checking hold time
Assume:

tsu = 0.6 ns
th = 0.4 ns
0.8 ns <= tcQ <= 1.0 ns
tgate = 1.0 + 0.1k

where
k = number of inputs

Shortest delay = tcQmin +  tNOT
= 0.8 + 1.1 = 1.9 ns

Since 1.9 ns > th = 0.4 ns,
no hold violation. 



Figure 5.67.   A 4-bit 
counter critical path.

Tmin = tcQmax +  3 tAND +  tXOR + tsu

= 1.0 +  3(1.2) + 1.2 + 0.6 = 6.4 ns
Fmax = 1/6.4 ns = 156.25 MHz.



Figure 5.25.   A modulo-6 counter with synchronous reset.

Figure 5.26.   A modulo-6 counter with asynchronous reset.

Figure 5.27.   A two-digit BCD counter.



Figure 5.28.   
Ring counter.



Figure 5.29.   Johnson counter.

Produces a sequence of length 2n.
0000  1000  1100  1110  1111  0111  etc.

module Dflipflop( input clock,

reset, D, output reg Q );

always @( posedge clock )

Q <= reset ? 0 : D;

endmodule

Code for a D flip-flop using a <= non-blocking assignment.



module Dflipflop( input clock,

reset, D, output reg Q );

always @( posedge clock )

Q <= reset ? 0 : D;

endmodule

Code for a D flip-flop using a <= non-blocking assignment.

The RHS of the <= 
operator is evaluated 
just before the clock 
edge and the 
assignment is made just 
after the clock edge.



module Dflipflop( input clock,

reset, D, output reg Q );

always @( posedge clock )

Q <= reset ? 0 : D;

endmodule



module JK( input clock, J, K, reset,
output reg Q );

always @( posedge reset,
posedge clock )

casex ( { reset, J, K } )
'b1xx: Q <= 0;
'b000: Q <= Q;
'b001: Q <= 0;
'b010: Q <= 1;
'b011: Q <= ~Q;

endcase

endmodule



Figure 5.36.  Incorrect code for two cascaded flip-flops.

Should use non-blocking 
assignments with clocked 
always! 

module IncorrectShiftRegister(
input clock, D,
output reg Q1, Q2 );

// Incorrect code.
// Both Q1 and Q2 always get
// the same value.

always @( posedge clock )
begin
Q1 = D;
Q2 = Q1;
end

endmodule



Figure 5.38.  Correct code for two cascaded flip-flops.

module CorrectShiftRegister( input clock, D,
output reg Q1, Q2 );

// Correct code.
// tCQ after the clock edge, Q1 and Q2
// take the values D and Q1 had tsetup
// before the clock.

always @( posedge clock )
begin
Q1 <= D;
Q2 <= Q1;
end

endmodule



Figure 5.40.  Another example of incorrect code

module BadCode( input clock, x1, x2, x3,
output reg f, g );

always @( posedge clock )
begin
f = x1 & x2;
g = f | x3;
end

endmodule



Figure 5.42. Correct code for Example 5.6.

module CorrectCode( input clock, x1, x2, x3,
output reg f, g );

always @( posedge clock )
begin
f <= x1 & x2;
g <= f | x3;
end

endmodule



module SynchReset( input clock, reset, D, output Q );

// Synchronous reset (synchronized to the clock)
always @( posedge clock )

Q <= reset ? 0 : D;

endmodule

module AsyncReset( input clock, reset, D, output Q );

// Asynchronous reset (not synchronized to the clock)
always @( posedge reset, posedge clock )

Q <= reset ? 0 : D;

endmodule

D flip-flops with synchronous and asynchronous resets.



module CounterA(
input clock, reset,
input [ 31:0 ] resetValue,
output reg [ 31:0 ] count = 0 );

// Synchronous reset (synchronized to the clock)
always @( posedge clock )

count <= reset ? resetValue : count + 1;

endmodule

module CounterB(
input clock, reset,
input [ 31:0 ] resetValue,
output reg [ 31:0 ] count = 0 );

// Asynchronous reset (not synchronized to the clock)
always @( posedge reset, posedge clock )

count <= reset ? resetValue : count + 1;

endmodule



CounterA CounterB 
always @( posedge clock ) 
   count <= reset ? 
      resetValue : count + 1; 

always @( posedge reset, posedge clock ) 
   count <= reset ? 
      resetValue : count + 1; 

 

From simulation, reset in synchronous in CounterA, changing only with the clock,  
and asynchronous in CounterB.


		CounterA

		CounterB



		always @( posedge clock )

   count <= reset ?
      resetValue : count + 1;

		always @( posedge reset, posedge clock )

   count <= reset ?
      resetValue : count + 1;









Figure 5.47.  Code for a D flip-flop with a 2-to-1 
multiplexer on the D input.

module MuxDFF1( input clock, s, D0, D1,
output reg Q );

always @( posedge clock )
if ( s )

Q <= D1;
else

Q <= D0;

endmodule



Figure 5.48.  Alternative code for a D flip-flop 
with a 2-to-1 multiplexer on the D input.

module MuxDFF2( input clock, s, D0, D1,
output reg Q );

always @( posedge clock )
Q <= s ? D1 : D0;

endmodule



module MuxDFF3( input clock, s, D0, D1,
output reg Q );

wire D;
assign D = s ? D1 : D0;

always @( posedge clock )
Q <= D;

endmodule

Yet another alternative code for a D flip-flop 
with a 2-to-1 multiplexer on the D input.



Figure 5.49.  Hierarchical code for a four-bit shift register.

module FourBitRightShift1( input clock, reset, D,
input [ 0:3 ] resetValue, output [ 0:3 ] Q );

MuxDFF1 ff0( clock, s, D,      resetValue[ 0 ], Q[ 0 ] );
MuxDFF1 ff1( clock, s, Q[ 0 ], resetValue[ 1 ], Q[ 1 ] );
MuxDFF1 ff2( clock, s, Q[ 1 ], resetValue[ 2 ], Q[ 2 ] );
MuxDFF1 ff3( clock, s, Q[ 2 ], resetValue[ 2 ], Q[ 3 ] );

endmodule



module FourBitRightShift2( input clock, reset, D,
input [ 0:3 ] resetValue, output reg [ 0:3 ] Q );

always @( posedge clock )
if ( reset )

Q <= resetValue;
else

begin
Q[ 0 ] <= D;
Q[ 1 ] <= Q[ 0 ];
Q[ 2 ] <= Q[ 1 ];
Q[ 3 ] <= Q[ 2 ];
end

endmodule

Another four-bit shift register.



module FourBitShiftReg3( input clock, reset, D,
input [ 0:3 ] resetValue, output reg [ 0:3 ] Q );

always @( posedge clock )
if ( reset )

Q <= resetValue;
else

Q <= { D, Q[ 0:2 ] };

endmodule

Another four-bit shift register.



module NBitRightShift1 #( parameter n = 4 )
( input clock, reset, D,
output reg [0 : n - 1] Q );

always @( posedge clock )
if ( reset )

Q <= 0;
else

begin
integer i;
for ( i = n - 1; i != 0; i = i - 1 )

Q[ i ] <= Q[ i – 1 ];
Q[ 0 ] <= D;
end

endmodule

An n-bit right shift register.



module NBitRightShift2 #( parameter n = 4 )
( input clock, reset, D,
output reg [ 0 : n – 1 ] Q );

always @( posedge clock )
if ( reset )

Q <= 0;
else

Q <= { D, Q[ 0 : n – 2 ] };
endmodule

An n-bit right shift register.



module UpCount( input clock, reset, load, enable,
input [ 3:0 ] loadValue, output reg [ 3:0 ] Q );

always @( posedge reset, posedge clock )
if ( reset )

Q <= 0;
else

if ( load )
Q <= loadValue;

else
if ( enable )

Q <= Q + 1;

endmodule



module Scan( input CLOCK_50,
output [ 9:0 ] LEDR );

reg [ 31:0 ] counter;

assign LEDR = counter[ 31:22 ];

always @( posedge CLOCK_50 )
counter <= counter + 1;

endmodule



module Scan( input CLOCK_50,
output [ 9:0 ] LEDR );

reg  [ 31:0 ] counter;
reg  [  3:0 ] onehot;
wire [  1:0 ] columnNumber;

assign columnNumber = counter[ 26:25 ];
assign LEDR = { onehot, columnNumber };

always @( posedge CLOCK_50 )
counter <= counter + 1;

always @( * )
case ( columnNumber )

0: onehot = 'b1000;
1: onehot = 'b0100;
2: onehot = 'b0010;
3: onehot = 'b0001;

endcase
endmodule



module Scan( input CLOCK_50,
output [ 9:0 ] LEDR );

reg [ 31:0 ] counter;
reg [  3:0 ] onehot;
reg [  1:0 ] columnNumber;

parameter desiredFrequency = 1.0/3.0,
divisor = 50_000_000 / desiredFrequency;

assign LEDR = { onehot, columnNumber };

always @( posedge CLOCK_50 )
if ( counter == 0 )

begin
counter <= divisor;
columnNumber <= columnNumber + 1;
case ( columnNumber )

0: onehot <= 4'b1000;
1: onehot <= 4'b0100;
2: onehot <= 4'b0010;
3: onehot <= 4'b0001;

endcase
end

else
counter <= counter - 1;

endmodule



Finite state machines

Machines whose next state depends on 
the inputs and the previous state.
Called finite state machines because they 
have only a finite number of states.



Combinatorial
logic

State variables
(Flip-flops)

Combinatorial
logic

Inputs
Outputs

Clock

Generalized form of a sequential circuit

If the outputs depend on both the current state and the 
current inputs, it is called a  Mealy machine, named after 
George Mealy, who invented the concept in 1955.



Combinatorial

Inputs

Clock

Moore machine

If the outputs depend only on the current state, it is called a  
Moore machine.  (“Moore is less.”) It’s named after Edward 
Moore, who invented the concept in 1956.



Mealy machines require fewer states but if the inputs change asynchronously, 
the outputs can change asynchronously as well.

Moore machines require more state variables and the outputs are delayed 
by one clock.  But all the outputs are guaranteed to be synchronous.

Flip-flops
Outputs

Clock

Simplest example:  A counter

Reset



A B

CD

State diagram for a 
counter with 4 states.

At each clock, it 
always moves to the 
next state.

The arrows between 
states are called 
edges or transitions.

Reset



We pick any state assignments 
we like, though some could be 
better than others.

State table
Present state Next state

A B
B C
C D
D A

A B

CD

Reset

State-assigned table
Present state Next state

y2 y1 Y2 Y1
A 00 01
B 01 10
C 10 11
D 11 00



A B

CD

Reset

Y1 y1
0 1

y2 0 1 0
1 1 0

Y1 = y1'

Y2 y1
0 1

y2 0 0 1
1 1 0

Y2 = y1 ^ y2

State-assigned table
Present state Next state

y2 y1 Y2 Y1
A 00 01
B 01 10
C 10 11
D 11 00

Having picked the assignments, we 
can use Karnaugh maps to derive the 
equations for the next state variables.



Y1 y1
0 1

y2 0 1 0
1 1 0

Y1 = y1'

Y2 y1
0 1

y2 0 0 1
1 1 0

Y2 = y1 ^ y2

D Q

QD

Set

Rst
Clock

Reset*

y1

y2



Y1 y1
0 1

y2 0 1 0
1 1 0

Y1 = y1'

Y2 y1
0 1

y2 0 0 1
1 1 0

Y2 = y1 ^ y2

module Counter( input clock, reset,
output reg y1, y2 );

always @( posedge reset, posedge clock )
if ( reset )

begin
y1 <= 0;
y2 <= 0;
end

else
begin
y1 <= ~y1;
y2 <= y1 ^ y2;
end

endmodule

In Verilog, the <= “non-blocking” 
assignment means all the assignments 
happen synchronously at exit from the 
always block.



State-assigned table
Present state Next state

y2 y1 Y2 Y1
A 00 01
B 01 10
C 10 11
D 11 00

In Verilog, we would probably skip 
the Karnaugh maps write the code 
directly from the state-assigned 
table.

module Counter2( input clock, reset,
output reg [ 1:0 ] y );

always @( posedge reset, posedge clock )
y <= reset ? 0 : y + 1;

endmodule



State-assigned table
Present state Next state

y2 y1 Y2 Y1
A 00 01
B 01 10
C 10 11
D 11 00

module Counter3( input clock, reset,
output reg [ 1:0 ] y );

always @( posedge reset, posedge clock )
if ( reset )

y <= 0;
else

case ( y )
0:  y <= 1;
1:  y <= 2;
2:  y <= 3;
3:  y <= 0;

endcase
endmodule

If the cases were more complex or 
not in order, we might write it like 
this with a case statement.



State-assigned table
Present state Next state

y2 y1 Y2 Y1
A 00 01
B 01 10
C 10 11
D 11 00

Or we might parameterize the 
assignments. module Counter4( input clock, reset,

output reg [ 1:0 ] y );
parameter A = 0, B = 1, C = 2, D = 3;
always @( posedge reset, posedge clock )

if ( reset )
y <= 0;

else
case ( y )

A:  y <= B;
B:  y <= C;
C:  y <= D;
D:  y <= A; 

endcase
endmodule



Verilog makes it really easy to pick any assignments 
you like and walk from one arbitrary state to 
another.

Example: Create a 3-bit counter in Verilog that 
cycles through this sequence:  4, 7, 0, 3, 2, 6, 1, 5. 



Example: Create a 3-bit counter in Verilog that 
cycles through this sequence:  4, 7, 0, 3, 2, 6, 1, 5. 

module PseudoRandom( input clock,
output reg [ 2:0 ] Q );

always @( posedge clock )
case ( Q )

4:  Q <= 7; 
7:  Q <= 0; 
0:  Q <= 3; 
3:  Q <= 2; 
2:  Q <= 6; 
6:  Q <= 1; 
1:  Q <= 5; 
5:  Q <= 4;

endcase
endmodule

